Newsgroup: it.comp.programmare.win32
Titolo: [VB] Form con Microsoft Internet Transfer Control

Qualcuno  ha  della  documentazione  /  ha  mai  fatto  /  un  esempio
dell'invio di dati al server con il metodo POST utilizzando l'Internet
Transfer Control ?
Grazie anticipatamente.
\begin{avviso}
 
Attenzione: Leggersi lo header riguardo al mio e-mail!
Warning: read header for infos about my e-mail.
\end{avviso}
+----------------------------------------------------------------+
| Gian Uberto Lauri * "So, are you working on fixing that bug    |
| 
sa...@eng.it
      *  *now* or leaving it until later ?" - YES  |
+----------------------------------------------------------------+

 Transfer Control ?
Ti posto un esempio di invio SMS (incompleto, il resto e' il compitino
:^).  Ti serve un form, 3 textbox, 1 combo, 1 command button, 1 combo,
1  status bar  e ovviamente 1 controllo inet. Apri il codice e incolla
il  seguente progetto.  All'interno non e' implementato nessun tipo di
gestione  errori ma  mostra come utilizzare l'unico evento esposto dal
controllo (StateChanged()).
Il  mio consiglio comunque e' di stare lontano dal quel OCX se ne puoi
fare  a  meno. Quasi tutti i  programmatori di VB preferiscono di gran
lunga  utilizzare direttamente le API o crearsi una dll per gestire le
connessioni  in  multithreading  (VB  fa SCHIFO  da  questo  punto  di
vista....aspettando i threads objs di VB 7).
Ciao.
Il Michett@
--
Private Sub Command1_Click()
    Dim strURL As String
    Dim strInputData As String
    ' Set trasmission protocol
    Inet1.Protocol = icHTTP
    ' Remove any pending connection
    Inet1.Cancel
    ' Set URL to post data
    strURL = "
http://free.kataweb.it/freesms/freesms_preview.html
"
    ' Data to POST
    strInputData = "prefix=" & Combo1.Tex & _
                   "&destination=" & Text1.Text & _
                   "&message=" & Text2.Text
    ' Send data with post method
    Inet1.Execute strURL, "POST", strInputData, _
        "Content-Type: application/x-www-form-urlencoded"
End Sub
Private Sub Inet1_StateChanged(ByVal State As Integer)
    Dim vtData As Variant
    Select Case State
        Case icConnected
            SetStatus "Connected"
        Case icConnecting
            SetStatus "Connecting"
        Case icDisconnected
            SetStatus "Disconnected"
        Case icDisconnecting
            SetStatus "Disconnecting"
        Case icError
            SetStatus "Error: " & Inet1.ResponseInfo
        Case icReceivingResponse
            SetStatus "Receiving response"
        Case icRequesting
            SetStatus "Requesting"
        Case icRequestSent
            SetStatus "Request Sent"
        Case icResolvingHost
            SetStatus "Resolving host"
        Case icResponseCompleted
            SetStatus "Response completed"
            Dim strData As String: strData = ""
            Dim bDone As Boolean: bDone = False
            ' Get first chunk.
            vtData = Inet1.GetChunk(1024, icString)
            DoEvents
            Do While Not bDone
                strData = strData & vtData
                ' Get next chunk.
                vtData = Inet1.GetChunk(1024, icString)
                DoEvents
                If Len(vtData) = 0 Then
                    bDone = True
                End If
            Loop
            Text3.Text = strData
        Case icResponseReceived
            SetStatus "Response received"
    End Select
End Sub
Private Sub SetStatus(strMsg As String)
    StatusBar1.Panels(1).Text = strMsg
End Sub
--
Inserisci nella propieta' list di combo1 i seguenti dati:
TIM
0333
0335
0338
0339
OMNI
0340
0347
0348
0349
WIND
0328
0329
BLU
0380

 writes:
IM>     strInputData = "prefix=" & Combo1.Tex & _
IM>                    "&destination=" & Text1.Text & _
IM>                    "&message=" & Text2.Text
Grazie,  avevo giusto  bisogno di  vedere  una riga  come questa.  Ora
vediamo  se crea  una  Query String  oppure  mi permette  di avere  un
payload più abbondante...
